

J Primer

作者: Eric Iverson 翻译: Taigua

目录

1	从这里开始	4
2	为什么学J?	5
3	本书的目标	7
4	需要的背景知识	8
5	如何使用本书	9
6	开始	10
7	实验	11
8	标准配置	13
9	术语	14
10	字母表	15
11	单词	17
12	句子	18
13	动词	19
14	名词	20
15	数字	21

16 负数	23
17 原语	25
18 名字	26
19 注释	28
20 错误	29
21 两可性	30
22 双子	31
23 单子	32
24 NuVoc - 新词汇表	33
25 第一个检查点	36
26 数值常量	37
27 字符串	38
28 单词构成	40
29 空格	43
30 优先级	46
31 括号	47
32 求值顺序	48
33 动词定义	51
34 单子/双子定义	54
35 脚本	57
36 局部名字	62

37 全局名字	64
38 调试全局名字	68
39 =. 与=:何时类似	70
40 =. 与=:何时不同	71
41 区域	72
42 z区域	76
43 脚本加载	78
44 第二个检查点	80

第 1 章

从这里开始

J是一种通用的高级编程语言。如果您是J的新手，并且想成为一名J程序员，那么这是一个很好的起点。即使您有相当丰富的编程经验，J也有很多独特之处，在深入研究之前，至少浏览一下这些内容是值得的。

第 2 章

为什么学J?

J是一种非常丰富的语言。你可能学习和使用它很多年，但仍然认为自己是初学者。这与Basic或Java等更简单的语言形成鲜明对比，在这些语言中，数月的努力学习和使用将使您成为专家。成为一名专家J程序员所需要的努力与成为一名专家c++程序员所需要的努力更接近。

好消息是，J的本质是如此简单和一致，你可以很快地学会解决真正有趣的问题。

学习足够的Basic或Java来解决琐碎的问题更容易，但学习足够的J来解决更有趣和更具挑战性的问题更容易。一旦你掌握了这种水平的技能，你就不会走到路的尽头，而是可以继续提高，使自己成为一个更好、更强大的程序员。

J尤其擅长对阵列数据进行数学、统计和逻辑方面的分析。它是一个强大的工具，可以为老问题构建新的更好的解决方案，甚至可以更好地找到尚未完全理解的问题的解决方案。

作为一种通用编程语言，J系统还提供：

- 两个集成的开发环境。JHS和JQt
- 标准库、实用程序和包
- 为应用程序构建事件驱动图形用户界面（JQt）
- 几种与其他编程语言和应用程序接口的方法
- 快速应用程序原型和开发
- 免版税发布应用程序的运行时版本

如果你对具有挑战性的数据处理问题的编程解决方案感兴趣，那么在学习J
上投入的时间将是值得的。

第 3 章

本书的目标

[NuVoc](#)是J语言的权威和明确的规范。[J字典](#)也可以用来学习J，但正因为它简洁、完整且严谨地涵盖了这门语言，并且更侧重于复杂之处而非寻常之处，它确实把我们中的一些人吓跑了。

本书为初学者提供了一个更亲切、更温和的起点。它带着你沿着一条路径，以轻松的步骤逐步前进，直到你能够用J语言编写应用程序。在这个过程中，你会通过简化和具体的情境来接触J的所有关键概念。这本手册的目的是让你快速上手，达到能够使用NuVoc和J字典的程度，并且用起来会让你纳闷：当初为什么还要为这些简单的东西费心呢？

你应该能够相当快地完成入门，最后将成为入门级的J程序员。这样，你将比最有经验的Basic或Java程序员拥有更强大的编程能力。

第 4 章

需要的背景知识

本书假设你熟悉另一种编程语言，如Basic、Java或C。然而，这不是先决条件，即使J是你的第一门计算机语言，应该也不会有什么特别的问题（实际上，更应祝贺你！）。

大多数事情都可以在J中完成，就像在其他语言中完成一样，并且通常一个主题的引入方式和在其他语言中引入它一样，这使得在J中更容易理解它是如何工作的。在某些情况下，有一种更好的J方法来解决这个问题，也会介绍这种方法。

第 5 章

如何使用本书

本书是一系列小的，易消化的部分，适合从头到尾按顺序阅读。一个章节通常需要你读过大部分甚至所有之前的章节。跳来跳去毫无意义，而且可能令人沮丧。

本书是自包含的，可以在不访问J系统的情况下读取。特别是，与J系统交互的示例显示了输入的内容以及系统如何响应。然而，它的目的是在访问系统和尽可能多地使用系统的情况下读取。强烈建议你最后输入所有示例，并尽可能多地把玩它们的变体。

有时章节使用的术语和概念直到后面才定义。这就要求你在第一次阅读时有一个初步的理解并继续，在第二次阅读时它们将变得更加具体。

本书最好读三遍：

- 浏览整本书。可以尝试一些例子，但最好持续前进，了解全局。
- 第二次仔细阅读并尝试所有的例子。
- 第三次尝试自己的例子来澄清你的理解，并提升你实际使用系统机制的舒适度（而不仅仅是阅读它和遵循说明）。

第 6 章

开始

本章假设你在[JQt IDE](#)（交互式开发环境）中使用[最新版本的J](#)。

Term窗口是一个执行窗口。你在Term窗口中键入J语句，当按Enter键时，J将执行它们并显示结果。

在Term窗口中键入以下行并按Enter键。

```
2 + 3
```

语句被执行并显示结果。输入以下行并按Enter键。

```
5 - 3
```

你的会话应该看起来像这样：

```
2 + 3
```

```
5
```

```
5 - 3
```

```
2
```

你输入的行缩进三个空格，J的答案从行首开始。

第 7 章

实验

我们鼓励你去尝试。尝试用不同的数字输入类似的行。很明显，+是加法，-是减法。输入一些用*表示乘，用%表示除的行。通过使用%，你将很快看到结果可以是2.5这样的数字，并且它们也可以用作参数。

在你有更多的经验之前，你有时可能会对你所观察到的东西感到惊讶甚至不安。一步一步来。尝试一些你确信你已经知道答案的例子，并做实验来证实你的理解。如果一个结果让你太困惑，早期阶段不要花时间在它上面。

3 - 5

_2

_2，而不是-2，可能会让你困惑。别担心，一会儿就会解释的。

本书中的大多数示例会显示你应输入的内容（缩进三个空格），同时也会显示系统返回的结果。这意味着你可以轻松地通读本入门手册，无需实际操作系统即可查看结果。然而，真正学会的唯一途径，最终还是靠你亲自尝试这些示例，并用自己的想法进行实验。示例采用等宽字体显示，与它们在系统Term窗口中的呈现方式基本一致。部分示例会使用较大的字体，以便你更轻松地阅读并在系统中输入——这通常出现在你可能因对某些词汇不熟悉而输错内容，或者输入错误可能导致令人困惑的结果时。

在做实验的时候，你经常想对你已经输入过的句子做一些细微的改变。有几个捷径可以使这更容易。在Term窗口中，你可以将光标移动到窗口中

的任意行，然后按Enter键，将该行作为窗口底部的新行，以便进行编辑。通过按住Ctrl+Shift并按向上箭头键，可以调出以前的输入行进行编辑，持续按向上箭头直到看到想要处理的行。

大多数小节中的示例都是独立的，但小节的后面部分可能依赖于前面部分中采取的步骤。有些小节依赖于前面几节中采取的步骤，但这应该是相当明显的。

第 8 章

标准配置

本书中的示例假设系统在JQt IDE中启动时运行标准配置文件。这将配置你的系统并提供一些标准实用程序。

在Term窗口中输入CR（大写字母C后跟大写字母R），以快速检查配置文件是否已运行。

```
CR
```

如果运行结果是空行，那么标准配置文件已经运行，你可以跳过本节的其余部分。

相反，如果你看到：

```
CR
```

```
|value error
```

那么你必须更改系统，以便运行标准配置文件。一个正常的J安装总是被配置为运行标准配置。 参考[安装J](#)。

第 9 章

术语

所有的编程语言都与英语有共同之处。在类比含义接近的情况下，J编程语言倾向于使用英语术语，而不是数学和其他编程语言中使用的术语。

像在其他语言中一样，你可以说“代码行”，但在J语言中，倾向于说“句子”。类似地，你可以把+称为“函数”，但在J中通常称为“动词”。

J使用的英语术语有：字母表、单词、句子、短语、动词、名词、副词和连词。

采用这种方法有几个原因。它解决的一个问题是：传统术语在数学和许多编程语言中有大量相关但又微妙不同的用法。例如：函数、子函数、运算符、程序、例程和子例程在不同的编程语言中都以略有不同的方式使用。J没有继承这种混乱，而是采用自己的术语，并在其上下文中精确地定义它们。

使用英语术语可以让你很好地了解术语在J中的一般含义。此外，使用自然语言术语可以鼓励和促进对问题进行英语陈述，并更直接地编写相应的J句子。

我们鼓励使用J术语，但肯定不是强制性的，使用术语“函数”而不是“动词”是完全可以接受的。

第 10 章

字母表

J编程语言的字母表是ASCII字母表，由以下部分组成：

26 lowercase letters (a to z)

26 uppercase letters (A to Z)

0 1 2 3 4 5 6 7 8 9

= < > _

+ * - %

^ \$ ~ |

. : , ;

! / \

[] { }

" ` @ & ?

()

,

有几个字符有时会引起混淆。-（减号）字符不同于_（下划线）字符，并且有三种不同的引号字符：

' quote

" double-quote

` back-quote

如果你尝试一个例子，输入"（双引号）误输入为''（两个单引号），你会感到失望，因为你的结果与书中的不一样。

句号.通常称为点。

第 11 章

单词

单词是字母表中有意义的一组字符。

2.5 + 5

这个句子有三个单词：数字2.5，+和数字5。

用字符组成句子中的单词的规则很简单，目前，常识足以识别句子中的单词。有一些复杂情况在后面的章节中讨论。

第 12 章

句子

句子是由一组单词组成的完整指令。

下面的2加5指令就是一个J语言的句子。

`2 + 5`

与英语句子不同，J语言的句子不以句号或其他标点符号结尾。

相反，J编程语言的句子通常是一个完整的行。

第 13 章

动词

在下面的句子中+是动词（表示动作的单词）。

2 + 5

第 14 章

名词

在下面的句子中，数字2和5都是名词。

$$2 + 5$$

第 15 章

数字

以下这些是数字：0 12 02.5 12.75 0.5 7e6

输入这些数字或其他数字，用动词+ - * %组成简单句子。

2 * 12.75

25.5

一个重要的规则是数字不能以点（.）开头。

0.5

0.5

0.5 + 3

3.5

.5

+--+--+

|.|5|

+--+--+

`.5 + 3`

| `syntax error`

| `.5+3`

显然，`.5`和`0.5`不一样。你现在不需要知道`.5`是什么，但重要的是明白数字不是以点开头的。

`_`（下划线）是无穷大，是一个数字。

第 16 章

负数

负数用_（下划线）表示，而不是-（减号）。

5 + _3

2

_7

_7

_5e_3

_0.005

5 - 9

_4

去掉-和9之间的空白并不会改变含义。

5 -9

_4

-总是动词。这简化了识别句子的规则，因为没有把-直接用在数字左边这种情况。代价就是，既然-总是一个动词，则需要另一个字符_（下划线）来拼写负数。

_用于拼写数字，放在有0 1 2 3 4 5 6 7 8 9，. 以及表示指数符号的e组成的数字前；表示一个负数。

记住：数字前面的 $\cdot_ \cdot$ 是数字的一部分，表示它是负数，而-总是一个动词，不是数字的一部分。

--（两个下划线）表示负无穷，是一个数字。

第 17 章

原语

原语是由系统定义的一个单词。例如，+是一个原语。原语的含义是固定的，不能改变。

原语可以由一个图形字符（如+）拼写的，或者用图形字符加上一个表示变形的字符（点号或冒号）拼写，如+.或者+:

原语也可以由一个或多个字母后跟一个点号或冒号拼写。例如，i.是一个原语，根据它的使用方式的不同，被称为**生成索引**或**求索引**。

第 18 章

名字

原语是由系统定义的单词，而名称是由你定义的单词。原语=.定义一个名字。

5 + v

28

这个句子可以读作v是23。单词=.叫做**系动词**（另一个很好的英语术语）。名字v被定义为数字23，可以用在其他句子中。

与原语不同，名字可以重新定义。

v =. 45

5 + v

50

当句子以系动词开始时，系统不会显示句子的结果。只包含名字的句子将显示名字所定义内容的显示形式。

abc =. 123

abc

123

你可以给任何东西起名字。例如，你可以给动词+起一个名字。

```
plus =. +
```

```
23 plus 45
```

68

`abc =. def`的一般读作：`abc`被定义为`def`。然而，借用其他计算机语言，也常说：`abc`被赋值为`def`或将值`def`赋给`abc`。

第 19 章

注释

原语NB.开始一个注释，该注释一直到行尾结束。

```
2 + 23  NB. this is a comment and is not executed
```

25

第 20 章

错误

句子中的错误将会被报告并停止执行。

```
123 foo 234
```

```
|value error: foo
```

```
| 123      foo 234
```

一条竖线标记了错误报告。如果可以的话，句子会用额外的空格来标记检测到错误的地方。在J9.4版本中，错误消息得到了改进，使调试更加容易。

第 21 章

两可性

每个动词都有两种定义：当它同时有左参数和右参数时使用双子（**dyad**），当它只有右参数时使用单子（**monad**）。这两个定义通常是相关的，比如-的双子定义表示减法，而单子定义表示求负。

5 - 3

2

- 7

_7

双子也被称为动词的二元模式，而单子被称为动词的一元模式。“两可性”一词用于化学意义上的两种价态，表明一个动词既可以与一个参数反应，也可以与两个参数反应。

记住：绝大多数动词都有单子和双子两种定义。

第 22 章

双子

如果一个动词同时有左参数和右参数，则使用动词的二元模式。

双子%（除以）被定义为左参数除以右参数。

5 % 2

2.5

第 23 章

单子

当动词只有一个右参数时，使用动词的一元模式。

单子%（倒数）被定义为1除以右边的参数。

% 2

0.5

单子%和双子%之间的关系很常见，其中单子是带有固定左参数的双子，你会在许多其他原语动词中看到这种关系。

第 24 章

NuVoc – 新词汇表

[NuVoc](#)（菜单Help|Vocabulary）是最新的J参考，是你关于J的最终、权威信息来源。你越早熟悉使用它越好，一个好的开始方法是查找到目前为止介绍的原语。

在JQt IDE中，按F1显示NuVoc，并单击原语跳转到条目。如果你按下F1同时按住Ctrl（Mac的Cmd），你会得到上下文敏感的帮助。如果光标在+处，按Ctrl+F1将跳转到原语+的条目。

在NuVoc中，查看包含+的第6行。这一行的第一项包含：+ 0 Conjugate . Plus 0 0。单词+是动词。它的单子称为求共轭，它的双子称为加法，就像算术中定义的那样。

单子+很有趣。在数学中，复数的共轭是实部相同，虚部相反的数。在实数上求共轭没有作用，结果是参数本身，在复数上它改变虚部的符号。J支持复数，就像直接支持整数或实数一样。复数用j分隔实部和虚部来表示。

```
int =. 23
```

```
+ int
```

```
23
```

```
float =. 23.5
```

+ float

23.5

imagine =. 2j3 *NB. 2 real part, 3 imaginary part*

+ imagine *NB. change sign of imaginary part*

2j_3

许多原语支持复数，NuVoc必须对此进行记录，这意味着在某些描述中存在一些额外的复杂性。如果你的应用程序中需要复数，这是非常棒的。但如果你是一个初学者，不关心复数，那么你必须知道足够的知识，从而能够忽略掉这些，而不会分心或困惑。

让我们看一下wiki中的Plus条目。你可以通过点击Plus链接访问。这将带你到描述+的页面中间。 wiki页面的上半部分是描述Conjugate的信息。Plus的标题行显示它是一个带参数x和y的双子动词。 下面是一个指向Rank 0 0的链接，以及一个指向为什么这很重要的链接。你很快就会知道的。

接着下面是+的简要描述。然后，该页面继续进行一般性讨论和示例。

回到NuVoc，看看+列右边的+.条目，它包含：+. Real/Imaginary . GCD (or)， 其中Real/Imaginary是双子，GCD (or)是单子。我们来看一下定义。

参考页面的上半部分涵盖了+.的Real/Imaginary用法。下半部分覆盖GCD (or)。

二元模式被描述为：GCD (Or)，它有两个非正式的名称，GCD和Or，表明它可以以两种不同的方式使用。 请注意，GCD代表最大公约数（这至少应该唤起数学记忆）。进一步在定义中，你将发现，如果参数是布尔值，则+.是逻辑or函数。 GCD是将or函数的域扩展到非布尔参数的有用扩展。这种原语域的扩展在J中很常见。现在有趣的是，请注意+.有这个较大的域，但也很容易把它限制到布尔参数上。

0 +. 0 *NB. 0 or 0*

0

```

0 +. 1
1
1 +. 0
1
1 +. 1
1

```

在NuVoc中再移动一列，我们可以看到词汇表条目+:包含: Double . Not-Or。该定义对单子和双子都给出了非常简单的定义。

一元模式的情况称为加倍，它的作用正如你所期望的那样。

```

+: 3
6

```

二元模式是实参逻辑或运算的否定。

同样，NuVoc中的一些一般性讨论和示例现在可能超出了你的能力范围。但关键是要知道如何导航并获取相关信息。

?和?.行之后，有一些其他原语行，这些原语是由用点或者冒号作为后缀的名字组成的。在原语表的下面是一个辅助页面列表，它们解释了J的一些概念。它们中的一些在你完成本入门之后值得一看，因为它们提供了你将看到的一些概念的更详细的信息。

第 25 章

第一个检查点

到了这个检查点，你应该明白：

- 如何使用J字典词汇表
- 术语：单词、句子、名词、动词、矛盾词、双子、单子等

通过做以下练习来检查你的理解：

- 查找单子+：*：-：%：的定义
- 试试这些新单子

第 26 章

数值常量

你已经看到了单个数字的使用。也可以有一个数字列表。

```
num =. 5
```

```
nums =. 23 0.5 12.5 7e6 _12 7
```

你将会对数字列表做更多操作，但现在我们只想建立这样一个列表。

第 27 章

字符串

用单引号括起来的字符创建一个字符串。

```
char =. 'Q'
```

```
chars =. 'this is a list of characters'
```

显示字符串时不显示单引号。

```
char
```

```
Q
```

```
chars
```

```
this is a list of characters
```

单引号用于字符串的开始和结束。一对单引号表示引号字符本身在字符串中。

```
'put 2 ''s to get 1 '' in the string'
```

```
put 2 's to get 1 ' in the string
```

不匹配的单引号是错误的。

```
abc =. 'asdf
```

```
|open quote
```

字符串可以为空。

```
abc =. ''
```

字符串也被称为字面值常量或字符常量。

第 28 章

单词构成

单词::可以帮助你找出句子中的单词。单词构成原语以字符串作为其右参数，将其分割成单词，并返回结果，每个单词都在一个盒子中。先不要担心这些盒子是什么，只需注意它们在视觉上的帮助。你将在后面的部分中了解盒子。

```
单词:: '2 + 3'
```

```
+-----+
```

```
|2|+|3|
```

```
+-----+
```

```
单词:: '2.5 + 3e4'
```

```
+---+---+
```

```
|2.5|+|3e4|
```

```
+---+---+
```

```
单词:: 'a =. 1 2 3'
```

```
+---+-----+
```

```

|a|=.|1 2 3|

+--+---+-----+

      ;: 'test + 123 NB. this is a comment'

+-----+--+---+-----+

|test|+|123|NB. this is a comment/

+-----+--+---+-----+

      ;: 'def =. ''testing 1 2 3'''

+-----+--+---+-----+

|def|=.|'testing 1 2 3'|

+-----+--+---+-----+

```

请注意，以下都是J单词，每个单词都在自己的盒子里：

2.5

3e4

=.

1 2 3

test

NB. this is a comment

'testing 1 2 3'

你可能会感到惊讶，诸如1 2 3和'testing 1 2 3'之类的常量都是J单

词。 这是很重要的一点，理解这一点在阅读和书写J句子时是必要的。

如果你确实对一个J句子感到困惑（这是可能发生的），你可以做的一件事就是对它应用;:， 以确保你知道构成它的单词。然后你就可以关注这些单词的意思了。

在[NuVoc](#)中查找;:。;:的非正式名称单词构成。

第 29 章

空格

不需要空格来分隔原语和其他单词。另一方面，额外的空格也不会改变意思。

`2+3`

`5`

`2 + 3`

`5`

`2    +    3`

`5`

本书中的大多数示例在所有原语周围都有空格。这使得单个单词显而易见，让你专注于单词的意思，而不需要先弄清哪些是单词。

然而，大多数J程序员，随着他们变得越来越有经验，达到了可以轻松阅读句子中的单词的程度，而额外的空格变成了理解的麻烦和障碍，而不是帮助。你将注意到，在后面的一些示例中，省略了一些不必要的空格。

在某些情况下，空间是必不可少的。

名字之间必须用空格分隔。

a=.3

plus=.

a plus a

6

aplusa

|value error: aplusa

而.（点）和:（冒号）用作单词屈折变化，将紧接在它们前面的单词变成一个新词。当它们用作原语或原语的开头时， 必须在前面有一个空格，这样它们就不会被视为屈折。

;: 'a . b' *NB. 3 words*

+--+--+

|a|.|b|

+--+--+

;: 'a .b' *NB. same 3 words*

+--+--+

|a|.|b|

+--+--+

;: 'a. b' *NB. 2 words*

+---+--

|a.|b|

+--+--+

如上所示，.或者:如果不是用作屈折，则必须和它们之前的单词之间用空格分开。

数字，例如1e7，可以包含字母。事实上，在J中，有几个字母可以用于拼写数字。而紧跟着数字的字母将被当作数字拼写的一部分。

plus =. +

1plus 3

| ill-formed number

| 1plus 3

| ^

1 plus 3

4

必须用空格将数字与非数字组成部分的字母分开。

第 30 章

优先级

数学传统上认为乘法优先于加法。在数学课上，如果有人问你 $2 + 3 * 4$ 等于多少，你要知道答案是14。

如果只有几个函数和几个级别的优先级(除法|乘法|加法和减法)，这并不太令人困惑。但事情在诸如C这样的语言中会变得笨拙，因为有许多功能和优先级级别。

由于J中有大量的动词，很难定义优先级，更不用说在阅读或编写时试图记住它了。此外，能够命名事物意味着你还必须弄清楚如何处理这个句子：

```
2 plus 3 times 4
```

与传统数学以及大多数其他编程语言不同，J没有动词优先级。你可能需要一点时间先停止乘法，但整体的简化是值得的。

记住：**没有**动词优先级。

第 31 章

括号

数学和大多数编程语言（包括J语言）都使用括号来控制句子的求值。如果一个句子被完全括起来，那么大多数语言的求值顺序是相同的，并且不受动词优先级或任何其他规则的影响。

$$2 + (3 * 4)$$

14

$$(2 + 3) * 4$$

20

$$10 - (4 - 3)$$

9

$$(10 - 4) - 3$$

3

对这些答案没有任何异议。

第 32 章

求值顺序

如果省略括号，答案是什么？

$$10 - 4 - 3$$

9

J对句子求值为：

$$10 - (4 - 3)$$

9

大多数其他语言会将其求值为：

$$(10 - 4) - 3$$

3

在没有括号的情况下，J从右到左隐式地产生语句。其他语言从左到右产生它们。 一个更长的句子会让这看起来更清晰。

$$10 - 4 - 3 - 1$$

8

10 - (4 - (3 - 1)) *NB. J right-to-left*

8

((10 - 4) - 3) - 1 *NB. others left-to-right*

2

现在考虑一组一元动词。

- - - 4

_4

每个人都知道如何加括号，每个语言都是一样的。

- (- (- 4))

_4

分组是从右向左进行的，在这种情况下，其他语言和J一样。J总是从右向左插入括号，而其他语言对于不同的情况有不同的规则。

J的求值顺序是从右到左。大多数其他语言都是从左到右对二元动词求值，从右到左对一元动词求值；而求值顺序又会被所涉及的动词的相对优先级修改。

对于名词和动词，[J词典E章节](#)的J求值规则是：

执行从右向左进行，但当遇到右括号时，先执行它及其匹配的左括号所包含的段，用其结果替换整个段及括号。

我们应该注意到，虽然[NuVoc](#)是J信息的权威来源，但J字典中有很多有用的信息可能没有进入wiki。

在J中，除了名词和动词之外，还有一些你还没见过的东西，它们会让这个规则变得更加复杂。正是这些额外的类在很大程度上证明了J打破传统 并采用从右到左的求值是合理的。进一步引自[J词典E章节](#)：

这些规则的一个重要后果是，在没有括号的表达式中，任何动词的右参数都是其右侧整个短语的结果。

这是因为没有动词优先级以及从右到左的求值的结果。

没有动词优先级、从右到左的求值以及其他类的规则使总体求值规则变得简单，减少了对括号的需要，并且使有经验的J用户更容易阅读和编写句子。

阅读下面的句子，在你的脑海中对它们求值，并理解无优先级和从右到左求值是如何解释答案的。

$$2 * 4 + 5$$

18

$$2 + 4 * 5$$

22

$$2 - 4 - 5$$

3

$$8 \% 2 + 2$$

2

记住：没有动词优先级和从右到左的求值。

第 33 章

动词定义

编程的艺术不在于使用原语，而在于定义适合自己需求的动词。在定义自己的动词时，你通过扩展语言来构建解决特定问题集的应用程序。

我们假设一个华氏温度和摄氏温度转换的问题。你需要定义一个动词来完成这个任务。

下面是动词摄氏度的定义，它将把它的参数从华氏温度转换成摄氏温度。

```
centigrade =. 3 : 0

t1 =. y - 32

t2 =. t1 * 5

t3 =. t2 % 9

)
```

此时，你想要做一些有用事情，而不是处理由拼写错误引起的问题，因此请仔细抄写。在终端窗口输入第一行：

```
centigrade =. 3 : 0
```

按所示输入。在3和:之间必须有一个空格。3表示正在定义一个动词，0表

示定义在随后的输入行中。

当你输入上述行之后，光标位于左侧行首，并且没有像通常那样缩进三个空格。这表明系统正在等待你输入定义的其余部分。

输入示例中摄氏度定义里剩下的行。它们位于左边界处，因此看起来它们可能是由系统显示的输出，但实际上它们是定义动词所需的行输入。

最后一行仅包含)的最后一行，结束了这种特殊的定义输入模式。输入最后一行后，系统再次缩进三个空格，表示准备执行一个句子。

如果你输入的定义正确，你应该可以尝试使用你的新动词。

```
centigrade 32
0
centigrade _40
_40
centigrade 212
100
```

让我们看看它的定义并理解它是如何工作的。定义的第一句中的y是动词实参的名称。当你执行带有参数的动词时，第一行将从参数中减去32并定义t1。当第一行完成后，继续执行下一行，它将t2定义为t1乘以5的结果。继续执行到下一行，并将t3定义为t2除以9。没有更多的行了，所以动词的执行完成了。动词的结果是最后计算的结果。

我们用3:0来定义动词。短语verb define和它是等价的，有些人觉得短语更容易阅读。事实上，如果你在系统中键入verb或define，你将看到它们已经分别被定义为3和:0。但是，短语隐藏了信息，而我们将使用3 : 0这种形式。

```
centigrade =. verb define
t1 =. y - 32
t2 =. t1 * 5
```

```
t3 =. t2 % 9
)
verb
3
define
:0
```

第 34 章

单子/双子定义

正如前面关于两可性的部分所讨论的那样，所有动词都有两种定义，单子和双子。你只定义了摄氏度的单子。那么双子呢？

```
23 centigrade 32

|domain error: centigrade

| 23 centigrade 32
```

由于你没有提供一个双子定义，它是空的，这被视为双子在其域中没有参数，你给出的任何参数都会导致域错误。

让我们看一些简单的例子来定义双子、单子和都定义的情况。

```
monadminus =. 3 : 0

- y

)

monadminus 5

_5

5 monadminus 3
```

```
|domain error: monadminus
```

```
| 5 monadminus 3
```

上面的代码定义了名为monadminus的动词的单子。一元应用有效，二元应用失败。

对于这种只有一行的定义，你可以采用快捷方式。将定义放在单行上，并避免进入需要以)结尾的特殊输入模式。 以下是完成上述定义的等效方法：

```
monadminus =. 3 : '- y'
```

到目前为止，你只定义了动词的单子。你也可以通过只定义双子来定义一个动词。使用4来定义双子，而不是将3作为:的左参数。

```
dyadminus =. 4 : 'x - y'
```

```
5 dyadminus 3
```

```
2
```

```
dyadminus 5
```

```
|domain error: dyadminus
```

```
| dyadminus 5
```

在单子的情况下，y是右参数；在双子的情况下，x是左参数，y是右参数。

如果你想同时定义一个动词的单子和双子呢？

```
minus =. 3 : 0
```

```
- y
```

```
:
```


`x - y`

)

行:分隔了单子和双子的定义。

`3 minus 5`

`_2`

`5 minus 3`

`2`

`minus 5`

`_5`

第 35 章

脚本

当你关闭J时，所有名字的定义都将失效。你在Term窗口中执行的操作只影响当前会话，但不是永久性的。这在实验时很好，但是当你开始定义像摄氏度这样的动词时，会想要记录下这个定义，以便在另一个会话中使用它。

关闭J并重新启动它。

```
centigrade
```

```
|value error: centigrade
```

因为你已经重新启动了，在上一个会话中定义的摄氏度动词和所有其他名字都丢失了。

但至少原语还在那里！

```
2 + 5
```

```
7
```

正如你所想的那样，为了永久记录定义，要将它们保存在文件中。带有J句子和定义的文件称为脚本文件，你可以像编辑任何其他文本文件一样编辑它们。脚本文件的后缀通常是*.ijs*。

记住：脚本文件是存储定义的源文件。

尽管你可以使用任何文本编辑器来处理脚本文件，但是J系统提供了一个简单的编辑器，该编辑器通过集成在J系统的方式使其更加方便使用。

当使用JQt界面时，`File | New temp`菜单命令创建一个新的脚本文件和一个用于编辑它的窗口。当你点击它时，将看到你的J会话有一个Term窗口和一个新的Edit窗口来编辑.ij文件。

Edit窗口中，.ijs文件的名称显示在其标题栏。在Edit窗口中按下Enter键不会执行该行，它只是将光标移动到新行的开头。

在Edit窗口中输入你的摄氏度定义。

```
centigrade =: 3 : 0
```

```
t1 =. y - 32
```

```
t2 =. t1 * 5
```

```
t3 =. t2 % 9
```

```
)
```

在第一行一定要使用=:而不是=.:=:创建一个全局定义。如果使用=., 则是一个局部定义。这个重要的区别将在接下来的几页中解释。现在，如果你希望赋值具有预期的效果，只需确保使用=:创建文件中的定义。

到目前为止，你只是将在窗口中更改编辑。该文件尚未更改，动词仍未定义。你必须通过运行脚本来执行句子。

在Edit窗口中点击Run | Load Script以将更改保存到文件中，然后执行文件中的每个句子。这类似于在Term窗口中输入文件的内容，只是没有显示句子和结果。Term窗口中唯一显示的是系统生成的语句，它表示文件中的语句已被执行。这句话类似于：`load 'c:/j9.4-user/temp/1.ijs'`，取决于你运行的J的版本。如果报告了一个错误（在Term窗口中输出，左侧有一个竖线），那么你的脚本中有拼写错误。在Edit窗口中更正文本并再次运行。

脚本文件中的句子已经执行完毕，现在已经定义了摄氏度动词。在Term窗口尝试使用摄氏度。

```
centigrade 32
```

0

使用File | New temp创建的文件位于J临时目录中，并具有临时形式名称（带有.ij后缀）。但是，现在最好在J用户目录中使用更合适的名称重新保存它。点击File | Save As...，向上一个目录级别，然后到用户目录，并将文件名设置为cf.ij。Edit窗口标题栏中的文件名将更改为新名称。

关闭cf.ij窗口并删除你的摄氏度定义。你可以使用功能动词erase来删除名字的定义，其参数是要删除名字的字符串。结果为1表示删除成功。

```
erase 'centigrade'
```

1

```
centigrade 212
```

```
|value error: centigrade
```

```
|      centigrade 212
```

使用File | Recent打开cf.ij窗口，使用Run | Load Script加载并运行定义摄氏度的脚本。

```
centigrade 212
```

100

让我们在cf.ij窗口中添加一个华氏度的定义。在你的摄氏定义后输入以下内容。同样，一定要使用=:。

```
fahrenheit =: 3 : 0
```

```
t1 =. y * 9
```

```
t2 =. t1 % 5
```

```
t3 =. t2 + 32
```

)

使用Run | Load Script加载并运行cf.ijs脚本中的句子。

```
fahrenheit 0
```

32

```
fahrenheit 451
```

843.8

关闭J并重启。

```
centigrade
```

```
|value error: centigrade
```

```
fahrenheit
```

```
|value error: fahrenheit
```

使用File | Recent并选择你的cf.ijs文件。然后点击Run | Load Script，一个类似下面的行

```
load'c:/j9.4-user/cf.ijs'
```

出现在Term窗口，表示执行了文件中的句子，而你的动词现在已经定义。

```
centigrade 32
```

0

```
fahrenheit 100
```

212

在Term窗口中出现的以load开头的行实际上是导致文件中的句子被执行的句子。 菜单命令只是执行这个句子的一种快捷方式。该字符串是要运行的文件的完整路径名。 手动输入完整路径名时，可以将其缩短为相对路径名。

```
load '~/j9.4-user/cf.ijs'
```

现在检查一下你的动词是否定义好了。

使用File | Recent打开你的cf.ijs文件进行编辑。

如果脚本中有错误怎么办？让我们在脚本中添加一个故意的错误，看看会发生什么。在脚本末尾添加foo 123行，然后再次运行脚本。

```
load 'c:/j9.4-user/cf.ijs'

|value error: foo

|      foo 123

| [-13]
```

J系统报告错误并停止执行脚本中句子。错误报告末尾的数字是脚本中出现错误的行号。

从脚本中删除错误并再次运行它。

第 36 章

局部名字

动词摄氏温度在其定义中使用了名字t1、t2和t3，但如果你在动词之外引用它们，它们就没有定义。

```
centigrade 32
```

```
0
```

```
t1
```

```
|value error: t1
```

```
t1 =. 123
```

```
t1
```

```
123
```

```
centigrade 212
```

```
100
```

```
t1
```

```
123
```

在摄氏度定义内使用t1和在定义外使用t1没有冲突。动词摄氏度并没有在它自身之外定义一个t1，正如value error所表明的那样，给摄氏度的t1中设置一个值并不会改变它定义之外的t1的值。

摄氏度内部使用的t1是局部名字。局部名字仅存在于动词中。在摄氏度外面使用的t1是一个全局名字。在执行动词定义是遇到的由连词=.定义的名字，是局部名字。

第 37 章

全局名字

在动词执行之外定义的名字是全局名子。

上一节中，Term窗口中定义的t1是一个全局名子，与动词摄氏度内定义的t1完全不同。

让我们来做些实验。使用File | New temp创建一个临时脚本文件，并在其中输入以下定义：

```
fooa =: 3 : 0    NB. =: is important
```

```
zzz + y
```

```
)
```

使用Run | Load script运行脚本。Term窗口显示：

```
fooa 5
```

```
|value error: zzz
```

```
|      zzz+y
```

让我们定义全局zzz，看看会发生什么。在动词外定义它，无论我们使用=.还是=:，它都是全局的。

为了遵循最佳实践，我们将在Term窗口中使用=::

```
zzz =: 23 NB. define global zzz
```

```
fooa 5
```

28

动词fooa使用全局的zzz。动词可以使用全局名字。

编辑脚本以添加 “ ‘foob ‘，然后运行脚本。

```
foob =: 3 : 0
```

```
zzz =. 7
```

```
zzz + y
```

```
)
```

在Term窗口中:

```
foob 3
```

10

```
zzz
```

23

动词foob使用它的局部zzz而忽略全局zzz。因此，动词可以使用局部名字而忽略同名的全局名字。

如果你想定义一个全局名字呢？全局连词=:（带:词形变化的=）定义了一个全局名字。编辑脚本以添加fooc，然后运行脚本。

```
fooc =: 3 : 0
```

```
gw =: y
```

```
lz =. y
```

```
)
```

在Term窗口中:

```
fooc 3
```

```
3
```

```
gw
```

```
3
```

```
lz
```

```
|value error: lz
```

```
gw =: 24 NB. Using =: to assign a global value, although  
↪ assignment with =. outside of a verb would also be  
↪ global.
```

```
fooc 5
```

```
5
```

```
gw
```

```
5
```

用=:定义gw，则定义了全局名字。

一般来说，在动词中只定义局部名字而不定义全局名字是一种很好的做法。这是有时被称为函数式编程风格的重要组成部分。定义全局名字的动词被认为具有副作用，并且更有可能导致bug，并且使阅读应用程序以理解正在发生的事情变得更加困难。事实上，在J的最新版本中，当你定义一个同时使用同名全局名字和局部名字定义的动词并且运行它，将会触发一

个域错误。

第 38 章

调试全局名字

有时，在尝试调试或更好地理解动词时，查看其局部名字或其他中间结果的值是很有用的。一种快速的方法是在动词定义中添加一行来进行全局定义。

打开`cf.ijs`文件并向摄氏度定义添加一行，将全局`gt1`定义为`t1`。

```
centigrade =: 3 : 0    NB. =:

t1 =. y - 32

gt1 =: t1              NB. temp line for debugging info

t2 =. t1 * 5

t3 =. t2 % 9

)
```

运行脚本。在Term窗口中：

```
centigrade 124

51.1111

t1
```

```
|value error
```

```
gt1
```

92

在摄氏度动词完成执行后，你无法看到局部`t1`的值，但你可以在`gt1`中看到值的副本。

从脚本中删除该行，并运行脚本以重新定义不含调试行的摄氏度动词。

第 39 章

=. 与=:何时类似

你已经看到了=.和=:用于动词定义时的不同之处。

当你在Term窗口中执行句子时，你不会在动词内执行它们，此时=.和=:有相同的效果。

NB. In the Term window

```
a =. 123
```

```
a
```

```
123
```

```
a =: 234
```

```
a
```

```
234
```

在Term窗口中连词=.和=:是相同的，你使用哪个来定义名字并不重要，因为它们都定义了全局名字。=.更容易输入，也更容易被使用。但严格来说，在定义全局名字时，最好显式地使用=:

第 40 章

=. 与=:何时不同

你已经看到了=.和=:在Term窗口中是相同的。同时在动词定义里，它们是不同的。但重要的是要认识到在脚本中也存在差异。

当你运行脚本时，load句子在Term窗口中执行，动词load执行脚本中的句子。因此，脚本中的句子在load动词中执行。这意味着用=.定义的名字被定义为动词load的局部名字。如果想在脚本中定义一个全局名字，必须使用=:。这就是为什么在脚本cf.ijs中定义全局名字（如摄氏度和华氏度）的行必须使用=:的原因。如果使用=., 它们将成为load的局部名字，并在load执行完后立即消失。

始终考虑定义是全局的还是局部的，并据此正确使用=:和=.

第 41 章

区域

首先，请注意区域（`locale`）与局部（`local`）是一个非常不同的单词，尽管后者只少了一个字母。

区域是一组全局名字。可以有多个区域，因此可以有多个全局名字集。

一个区域中的全局名字与其他区域中的相同名字的区别是，通过添加用_（下划线）括起来的区域名称后缀来限定名字。带有区域限定的名字始终是全球名字。

```
abc_def_ =: 2
```

上面的句子可以读作在`def`区域中的全局`abc`是2。

```
abc_base_ =: 4
```

上面的句子可以读作在`base`区域中的全局`abc`是4。

如果省略了区域名称，则假定它是`base`区域的名字。

```
abc__ NB. the same as abc_base_
```

4

如果全局名字没有使用区域名称进行限定，则它位于当前区域中。`base`区域是当前默认区域，除非在不同的区域中执行动词显式地更改了当前区

域。下面的代码定义了base区域中的abc:

```
abc =. 6

abc_base_

6

abc

6
```

由于base区域是当前默认区域，因此名字abc和abc_base_是相同的。

名字abc_def_显然不同于abc，但到目前为止还没有办法知道发生了什么特别的事情。在什么意义上abc和foo在同一个（base）区域中？而abc和abc_def_在不同的区域？

一种区分方法是使用列出全局名字的工具动词names。

```
a =. 23

b =. 24

a_q_ =. 25

w_q_ =. 26

names 0 NB. 0 lists nouns

a abc b
```

你的names结果可能不同，但它将包括你在base区域中定义的所有全局名字。你应该看到上面定义的a和b，并注意到没有看到在q区域中定义的w。

要查看在q区域中定义的名字，你可以执行以下操作：

```
names_q_ 0 NB. names in locale q

a w
```

名词a和w在q区域中定义。

区域将全局名字划分为不同的集合，而诸如names之类的实用程序可以操作特定区域中的全局名字。

区域的真正威力在于区域中定义的动词。当动词在一个区域中执行时，它将该区域（而不是base区域）作为当前区域。

让我们在q区域中定义一个简单的动词，看看它是如何工作的。

```
f_q_ =. 3 : 'a =: y'
```

这个动词用右参数定义全局a。可以有許多不同的区域，每个区域都有自己的全局a。但是当f_q_执行时，它在q区域中执行，q区域是当前的区域，它使用的全局名字来自q区域。试试下面的实验：

```
a =. 23      NB. define a in the base locale
```

```
a_q_ =. 24    NB. define a in locale q
```

```
f_q_ 100 NB. execute f in locale q
```

```
100
```

```
a
```

```
23
```

```
a_q_
```

```
100
```

区域将全局名字划分为单独的集合。特别是，相关的名词和动词（比如在一组实用程序中），可以在它们自己的区域中定义。它们的名字不会与base区域或其他区域中的名字冲突。当你查看应用程序时，你可以只查看特定区域中的相关全局名字。当动词在区域中运行时，它使用来自同一区域的全局名字。

参数为0时，`names`动词列出名词，参数为3时列出动词，参数为6时列出区域名称。

区域名称列表很有趣。`base`和`q`大家都知道，但是`j`和`z`呢？

`j`和`z`区域中的全局名字是在J启动并运行`profile.ijs`脚本文件时定义的。

`j`区域包含在构建应用程序时很有用的东西，在[J在线文档](#)中有详细讨论。

`z`区域确实非常有趣。

第 42 章

z区域

z区域是所有其他区域的父区域。

如果在当前区域中找不到名字，但在z区域中有它的定义，则使用该定义，就好像它在当前区域中一样。

z区域用于你希望在任何地方都可用的公共实用程序。通过z区域，它们可以在任何区域中执行，就像它们在该区域中一样，但是只有一个副本，并且z区域中的名字不会混淆其他区域中的名字。

profile.ijs在启动J时运行，它在z区域中定义了许多标准实用程序。你已经使用了在z区域中定义的erase和names动词。你可以通过以下方式来确认：

```
names 3 NB. verbs in the base locale  
...
```

上面的代码没有将names作为名字列出，但是你可以执行它。这是因为当在base区域中找不到它时，它在z区域中的定义就会被使用，就好像它实际上是在base区域中定义的一样。

```
names_z_ 3 NB. verbs in z locale  
...
```

结果太长了，就不在这里列出了。动词**names**有一个二元定义，它接受一个左参数，该参数指示要返回的名字的第一个字母。

```
'n' names_z_ 3
```

```
nameclass namelist names nc nl ...
```

第 43 章

脚本加载

除了与标准配置文件一起加载的实用程序之外，系统还提供了几个额外的标准实用程序脚本。这些标准实用程序在[J在线文档](#)中有记录，可通过J帮助菜单获得。你可以直接运行这些脚本，但是你需要记住脚本的路径，以及在哪个区域中运行它们。标准配置文件提供了一些实用程序，使你更容易完成此类操作。`scripts`动词列出了可以使用`load`动词加载的脚本。

```
scripts ''  
  
...  
  
.... parts    plot      profile  scripts  stdlib ...  
  
...
```

带有`v`参数的`scripts`动词列出了脚本及其完整路径。

`~system`表示J系统目录。

`convert`脚本包含几个转换实用程序。

```
load 'convert'  
  
toupper 'testing 1 2 3'  
  
TESTING 1 2 3
```

```
tolower 'Sir Richard'
```

```
sir richard
```


第 44 章

第二个检查点

到了这个检查点，你应该明白：

- 作为句子来源的文本文件称为脚本文件
- 脚本文件定义全局名字
- 如何创建一个新的临时脚本文件
- 如何将临时脚本文件保存为永久文件
- 如何运行脚本文件来执行语句
- 如何在脚本文件中定义动词
- 如何定义一个动词的一元和二元模式
- `=.`和`=:`之间的区别
- 局部名字和全局名字的区别
- 区域是一组全局名字
- 可以有多个区域
- `base`区域是你通常使用的

通过做以下练习来检查你的理解：

- 创建一个新的临时脚本文件
- 在脚本中将`square`定义为使用`*`对其参数进行平方的单子
- 将脚本保存在用户目录中，并命名为`square.ijs`
- 运行脚本并测试动词`square`
- 关闭J，重新启动，使用`load '~/j9.4-user/square.ijs`运行`square.ijs`并测试它（路径描述可能因J版本而异）